
Preface

Purpose
This book introduces the Graph Template Language (GTL) and the statistical graphics (SG) proce-
dures SGPLOT, SGSCATTER, and SGPANEL. The GTL enables you to make a wide variety of
modern statistical graphs by using a powerful language. The SG procedures enable you to make a
more limited set of graphs by using a simpler but less powerful syntax.

Whenever possible, this book shows you how to make a graph two ways: by using an SG procedure
and by using the GTL. These examples are designed to help you get started using both the SG
procedures and the GTL. Understanding the simple GTL examples, even when they are harder
to program than their SG counterpart, enables you to understand how to use the GTL for more
complicated graphs and for modifying the templates used by the analytical procedures (such as those
in the SAS/STAT, SAS/QC, and SAS/ETS products).

Is This Book for You?
If you are a SAS user and you need to create graphs, then this book is for you.

Scope of This Book
This book illustrates the major statements in the GTL and the SG procedures. It shows how to make
a wide variety of statistical graphs. However, it does not discuss the underlying statistical methods,
data analysis, or interpretation of results.

Software in This Book
� The GTL and SG procedures are part of Base SAS software.
� Some of the examples use SAS/STAT software to prepare data sets for display.



Sample Code

The beginning of every example (or every chapter for the short chapters) has a link that you can
double-click to see the SAS code for that example. The examples in this preface are available from
here:

Code links work in Adobe Reader and Microsoft Internet Explorer but not in most other browsers.
The sample code is embedded in this PDF file, so you can download the PDF file and then use it
without Internet access. Some browsers cannot handle that. Code links might not be displayed if you
print this book. You can enable code link printing in Adobe Reader by selecting Edit I Preferences
I Commenting and checking Print notes and pop-ups before printing.

ODS Graphics
Effective graphics are indispensable for modern statistical analysis. They reveal patterns, differ-
ences, and uncertainty that are not readily apparent in tabular output. Graphics provoke questions
that stimulate deeper investigation, and they add visual clarity and rich content to reports and
presentations.

There are three major ways in which you can use statistical graphics:

� Exploring your data. Many statistical analyses begin by producing simple graphs of data,
such as scatter plots, histograms, and box plots.

� Analyzing your data. Graphical results are an integral part of the results of modern statistical
analyses.

� Presenting your data. The final results of statistical analyses are often presented by using
graphs. Presentation graphics might be the same as the graphical results of analyses, or they
might be customized in various ways.

The SAS System provides many tools for creating statistical graphics. In many cases, multiple tools
can be used to make a graph.

� Automatically created graphs. Statistical graphs are created by using an extension of the
Output Delivery System (ODS). ODS manages procedure output and lets you display it in a
variety of destinations, such as HTML and RTF. When you use ODS Statistical Graphics (ODS
Graphics for short), statistical procedures produce graphs as automatically as they produce
tables, and graphs and tables appear together in the ODS output. You only need to enable ODS
Graphics by using the ods graphics on statement to get default graphical output. Some
graphs also require you to specify one or more simple options.

� SG procedures. The SG (statistical graphics) procedures SGPLOT, SGSCATTER, and
SGPANEL provide a simple and convenient syntax for producing many types of statistical
graphs. They are particularly convenient for exploring and presenting data.

� GTL. The Graph Template Language and PROC SGRENDER provide a powerful syntax for
creating custom graphs. You can also modify the templates that SAS provides for use with
SAS/STAT procedures to create customized results.

ods graphics on;



proc glm data=sashelp.class;

   model weight = height;

quit;



proc sgplot data=sashelp.class;

   title 'Fit Plot for Weight';

   reg y=weight x=height / cli clm;

run;



proc template;

   define statgraph FitPlot;

      begingraph;

         entrytitle 'Fit Plot for Weight';

         layout overlay;

            modelband "cliband" / display=(outline)

                      outlineattrs=GraphPredictionLimits

                      name='cli' legendlabel='95% Prediction Limits'

                      datatransparency=0.5;

            modelband "clmband" /

                      fillattrs=GraphConfidence

                      name='clm' legendlabel='95% Confidence Limits'

                      datatransparency=0.5;

            scatterplot y=weight x=height;

            regressionplot y=weight x=height / name='reg'

                           clm="clmband" cli="cliband" legendlabel='Fit';

            discretelegend 'reg' 'clm' 'cli';

         endlayout;

      endgraph;

   end;

run;



proc sgrender data=sashelp.class template=FitPlot;

run;







Table 1 provides a summary of ODS Graphics functionality.

Table 1 Summary of ODS Graphics Functionality

Graphical task Audience What do you use? What should you learn?

4 Create graphs in the
context of statistical
analyses

Statistical
users

Statistical procedures
in SAS/STAT,
SAS/ETS, SAS/QC,
and Base SAS that
support ODS
Graphics

Specify ODS GRAPHICS
ON; graphs are then
created by default or by
procedure options, which
are documented in the
procedure chapters.

Enhance specific
graphs for a paper
or presentation

Statistical
and general
SAS users

ODS Graphics Editor How to request editable
graphs, invoke the editor,
use point-and-click
features; see SAS 9.4 ODS
Graphics Editor: User’s
Guide.

4 Create stand-alone
graphs for data
exploration or for
customized displays

Statistical
and general
SAS users

SGPLOT, SGPANEL,
SGSCATTER
procedures in
Base SAS

SG procedure syntax; see
SAS ODS Graphics:
Procedures Guide.

4 Change the overall
appearance of
graphs and tables

Statistical
and general
SAS users

ODS styles STYLE= option in ODS
destination statement

4 Save and manage
graphs for papers
and presentations

Statistical
and general
SAS users

ODS GRAPHICS
options, ODS
destination options

How to specify size and
resolution; how to name
and access image files

4 Make persistent
changes in graphs
produced by
statistical
procedures (apply
whenever you run
your program)

Advanced
SAS pro-
grammers

User modifications of
graph templates that
SAS provides

Basic features of the
Graph Template Language
and PROC TEMPLATE;
see SAS Graph Template
Language: User’s Guide.

Create a highly
customized
stand-alone graph

Advanced
SAS pro-
grammers

ODS Graphics
Designer

GUI for creating graph
templates

4 Create a highly
customized
stand-alone graph

Advanced
SAS pro-
grammers

User-written graph
templates

Graph Template Language,
PROC TEMPLATE, and
PROC SGRENDER; see
SAS Graph Template
Language: User’s Guide
and SAS Graph Template
Language: Reference.

4 Discussed in this book.

The graphs that SAS/STAT procedures automatically produce are suitable for use in reports. However,
you might want to customize them first (for example, by using the point-and-click ODS Graphics
Editor, as described in SAS 9.4 ODS Graphics Editor: User’s Guide). This book focuses primarily
on the GTL, its use in creating customized graphs, and the SG procedures. Both simple and complex
examples are provided to help you get started creating modern statistical graphics by using the GTL
and SG procedures.1

ODS Graphics is available in procedures in SAS/STAT, Base SAS, SAS/ETS, SAS/QC, SAS
Enterprise Miner, and other products. The following step creates the regression fit plot displayed in
Figure 1 along with the ANOVA table and fit statistics in Figure 2:

ods graphics on;

proc glm data=sashelp.class;
model weight = height;

quit;

Figure 1 Fit Plot Created by PROC GLM

Figure 2 ANOVA and Fit Statistics

Dependent Variable: WeightDependent Variable: Weight

Source DF
Sum of

Squares Mean Square F Value Pr > F

Model 1 7193.249119 7193.249119 57.08 <.0001

Error 17 2142.487723 126.028690

Corrected Total 18 9335.736842

1Emphasis is placed on the code and the graphs, not on the data or the results.

Figure 2 continued

Basic ODS Graphics functionality is discussed in many places, including in “A Primer on ODS
Statistical Graphics” (Chapter 21,SAS/STAT User's Guide). The documentation chapter for each
SAS/STAT, SAS/QC, Base SAS, and SAS/ETS procedure that uses ODS Graphics provides examples,
information about which graphs are produced by each procedure, and information about what syntax
(if any) is required to make each plot.

ODS Graphics provides more than just a way to automatically produce graphs from analytical
procedures. It also provides two powerful ways to produce custom graphs. First, there are the
SG procedures: PROC SGPLOT, PROC SGSCATTER, and PROC SGPANEL.2 They provide a
high-level syntax for producing scatter plots, histograms, bar charts, box plots, scatter plot matrices,
and many other types of statistical graphs. You can use these procedures instead of PROC GPLOT,
PROC GCHART, and other legacy SAS/GRAPH procedures. The SG procedures are designed
speci�cally for statistical work, and they have statistical computation facilities built directly into
them. For example, the following step produces a scatter plot that has a linear regression line,
con�dence limits, and prediction limits and is very similar to the plot that PROC GLM produces:

proc sgplot data=sashelp.class;
title �Fit Plot for Weight�;
reg y=weight x=height / cli clm;

run;

The results are displayed in Figure 3.

Figure 3 Fit Plot Created by PROC SGPLOT

2ODS Graphics does not need to be enabled by specifying the ODS GRAPHICS ON statement in order for you to use
the SG procedures or PROC SGRENDER.

The plots in Figure 1 and Figure 3 differ in the degree of transparency of the band that displays the
con�dence limits and in the legend entries.

In addition to the SG procedures, Base SAS provides you with the Graph Template Language, PROC
TEMPLATE, and PROC SGRENDER.3

The GTL is a powerful language for de�ning the layout and composition of a graph or a panel of
graphs. You can use it to make very simple and very complex graphs. It is the same language that
SAS procedures use to de�ne their ODS Graphics. SAS procedures that provide ODS Graphics use
a template, written in the GTL, for each graph. You too can use the GTL and PROC TEMPLATE to
create graph templates. Then you can use PROC SGRENDER to produce graphs from a SAS data
set and the instructions provided in the graph template. For example, the following step produces a
scatter plot that has a linear regression line, con�dence limits, and prediction limits and is identical
to the plot that PROC GLM produces:

proc template;
define statgraph FitPlot;

begingraph;
entrytitle �Fit Plot for Weight�;
layout overlay;

modelband "cliband" / display=(outline)
outlineattrs=GraphPredictionLimits
name=�cli� legendlabel=�95% Prediction Limits�
datatransparency=0.5;

modelband "clmband" /
fillattrs=GraphConfidence
name=�clm� legendlabel=�95% Confidence Limits�
datatransparency=0.5;

scatterplot y=weight x=height;
regressionplot y=weight x=height / name=�reg�

clm="clmband" cli="cliband" legendlabel=�Fit�;
discretelegend �reg� �clm� �cli�;

endlayout;
endgraph;

end;
run;

proc sgrender data=sashelp.class template=FitPlot;
run;

The results are displayed in Figure 4. The graph template statements and options that are used in this
example are explained in detail throughout the book.

3The procedure name “SGRENDER” begins with “SG”; however, the phrase “SG procedures” usually refers only to
the SGPLOT, SGSCATTER, and SGPANEL procedures.

Figure 4 Fit Plot Created with the GTL

These three examples illustrate an important point. In most cases, you can simply enable ODS
Graphics and use an analytical procedure to produce your graphs. When you need additional graphs
or customized graphs, you can often use an SG procedure to get what you need. However, you can
always use the GTL to specify precisely what you need. Using the GTL will almost always require
more work, but it gives you the most power. The SG procedures and ODS Graphics are documented
in SAS ODS Graphics: Procedures Guide, SAS Graph Template Language: User's Guide, andSAS
Graph Template Language: Reference.

This book is organized by type of graph. Alternative ways of creating the same graph are presented
together and in parallel. The book provides a gentle, parallel, and example-driven introduction to the
SG procedures and to the GTL. Most graphs are produced in at least two ways. One graph is created
by using the GTL. Whenever possible, the other graph is created more directly by using one of the
SG procedures. Some graphs are also produced by SAS/STAT procedures or by using additional
options. Each example provides prototype code for getting started using the SG procedures and the
GTL. Examples of all the basic graphs that can be produced by the SGPLOT, SGSCATTER, and
SGPANEL procedures are provided. Although you do not need to write a template to make many
useful graphs, understanding the GTL enables you to create custom graphs that the SG procedures
cannot produce. It also helps you modify the sometimes complex templates that SAS provides.

Examples of all the major statements in the GTL (excluding annotation statements) are provided.
These statements can be classi�ed as follows:

� Plot statements specify a number of commonly used displays, including scatter plots, his-
tograms, contour plots, surface plots, and box plots.

� Layout statements specify the arrangement of the components of the graph.

� Text statements specify the descriptions that accompany graph elements.

� Control statements specify the conditional or iterative �ow of control.

This book is about graphics, not statistics. Statistical graphics are created and displayed, but the book
does not talk about interpretation or the complexities of data analysis. It uses simple data sets, often
theSashelp.Class data set that is familiar to most SAS users. Other data sets are used as needed.
The goal is to show you how to make graphs, not explain why you would make them or what you
would do with them.

Role of the GTL in SAS

It is important to understand that the GTL is a language developed for internal use at SAS. SAS
procedure writers use it to make the graphs that are automatically produced by analytical procedures.
You can use the same GTL to make your graphs. The GTL was developed to be a comprehensive
language for capturing the de�nition of potentially very complex graphs. Although SAS makes this
language available to you, it is not designed to be an obvious step away from syntax that is familiar
to longtime SAS users. It is a new and different language. However, as you become experienced with
the GTL, you will �nd that it is not dif�cult to use. You will simply �nd that it differs from what you
might expect if you are an experienced SAS user. This is discussed in more detail in Appendix B,
“Some Tips and Techniques for Understanding the Graph Template Language.” If you �nd that your
templates are not working the way you think they should, you might �nd the answer in this appendix.

Navigating This Book

Much of this book assumes that you are already familiar with using ODS Graphics together with
SAS analytical procedures but are not familiar with the SG procedures or the GTL. If you are not
familiar with the ODS Graphics statements, styles, destinations, and so on, see “Introduction to
ODS Graphics” on page 181 for an introduction. You can �nd an even more detailed introduction in
Chapter 21, “Statistical Graphics Using ODS” (SAS/STAT User's Guide).

Appendix B, “Some Tips and Techniques for Understanding the Graph Template Language,” provides
you with some background information about the GTL. It is important that you understand these
concepts when you start writing your own templates. Sections in Chapter 1, “The Graph Template
Language and the SG Procedures,” illustrate ways to make different types of graphs and illustrate
one or more statements in the GTL and in the SG procedures. The common statements (DEFINE,
BEGINGRAPH, LAYOUT OVERLAY, and so on) are explained in detail only in the section “Scatter
Plot” on page 3. Subsequent examples assume that you have read this �rst example. You should also
read the section “Regression Fit Plot” on page 6 before proceeding to other examples. Other than

those sections, you do not need to read the sections in order. If you come across statements that are
unfamiliar, you can use the index to �nd earlier examples of them.

The following list can help you �nd the section that illustrates each type of plot:

� “Axis Tables” on page 64
� “Bar Chart” on page 49
� “Block Plot” on page 128
� “Box Plot” on page 88
� “Bubble Plot” on page 109
� “Continuous Legend” on page 122
� “Contour Plot” on page 119
� “Contour Plot and Scatter Plot Overlaid” on page 124
� “Data Panel” on page 159
� “Density Plot” on page 78
� “Dot Plot” on page 98
� “Drop Lines” on page 84
� “Ellipses” on page 104
� “Fringe Plot” on page 82
� “Grouped Regression Fit Plot” on page 16
� “Heat Maps” on page 37
� “Histogram” on page 73
� “Line Plot” on page 111
� “Loess Fit Plot” on page 11
� “Needle Plot” on page 114
� “Penalized B-Spline Fit Plot” on page 14
� “Polygon” on page 100
� “Regression Fit Plot” on page 6
� “Regression Fit Plot with Con�dence and Prediction Limits” on page 23
� “Residual Panel” on page 148
� “Scatter Plot” on page 3
� “Scatter Plot Matrix” on page 139
� “Series Plot” on page 95
� “Step Plot” on page 115
� “Three-Dimensional Histogram” on page 135
� “Three-Dimensional Surface Plot” on page 133
� “Text Insets and Special Characters” on page 32
� “Vector Plot” on page 117
� “Waterfall Chart” on page 131

Resources

SAS offers you a rich variety of resources to help build your SAS skills and to explore and apply the
full power of SAS software. Whether you are in a professional or academic setting, SAS provides
learning products that can help you maximize your investment in SAS.

Bookstore http://support.sas.com/publishing/
Training http://support.sas.com/training/
Certi�cation http://support.sas.com/certify/
Knowledge Base http://support.sas.com/resources/
Support http://support.sas.com/techsup/
Learning Center http://support.sas.com/learn/
Community http://support.sas.com/community/
Documentation http://support.sas.com/documentation/

Base SAS http://support.sas.com/documentation/onlinedoc/base/
ODS http://support.sas.com/documentation/onlinedoc/ods/
ODS Graphics http://support.sas.com/documentation/prod-p/grstat/
SAS/STAT http://support.sas.com/documentation/onlinedoc/stat/

Sanjay Matange
Author Page http://support.sas.com/publishing/authors/matange.html
Blog http://blogs.sas.com/content/graphicallyspeaking

SAS Press books about ODS Graphics include the following:

� Statistical Graphics Procedures by Example: Effective Graphs Using SASby Sanjay Matange
and Dan Heath

� Getting Started with the Graph Template Language in SAS: Examples, Tips, and Techniques
for Creating Custom Graphsby Sanjay Matange

� Clinical Graphs Using SASby Sanjay Matange

Typographical Conventions Used in This Book

This book uses several type styles for presenting information. The following list explains the meaning
of the typographical conventions used in this book:

roman is the standard type style used for most text.

UPPERCASE ROMAN is used for SAS options, and other SAS language elements when they
appear in the text. However, you can enter these elements in your own
SAS programs in lowercase, uppercase, or a mixture of the two.

MonoSpace is used for longer SAS options, statements, and statement fragments
when they appear in the text. Lowercase or mixed case is usually used,

but you can enter these elements in your own SAS programs in lowercase,
uppercase, or a mixture of the two.

oblique is used for user-supplied values for options in the syntax de�nitions.

helvetica is used for the names of variables and data sets when they appear in the
text.

monospace is used for example code. This book usually uses lowercase or mixed
case for SAS code.

About This Edition

This book is a substantial revision ofStatistical Graphics in SAS: An Introduction to the Graph
Template Language and the Statistical Graphics Procedures(last published in 2010) and is a
companion toAdvanced ODS Graphics Examples(2015). Both were written by the current author.
This revision contains many new statements that have been added to PROC SGPLOT. They are all
covered in Chapter 1, “The Graph Template Language and the SG Procedures.” Chapter 3, “Style
Templates,” and Chapter 4, “Graph Template Modi�cation,” are now shorter because these topics
are well covered in Chapter 21, “Statistical Graphics Using ODS” (SAS/STAT User's Guide), and
Chapter 22, “ODS Graphics Template Modi�cation” (SAS/STAT User's Guide). For more information
about ODS and ODS Graphics, see Chapter 20, “Using the Output Delivery System” (SAS/STAT
User's Guide), and Chapter 23, “Customizing the Kaplan-Meier Survival Plot” (SAS/STAT User's
Guide).

This book was written for the third maintenance release of SAS 9.4.

About the Author

Warren F. Kuhfeld is a Distinguished Research Statistician Developer in SAS/STAT Research and
Development at SAS. Warren received his PhD in psychometrics from the University of North
Carolina at Chapel Hill in 1985 and joined SAS in 1987. He has used SAS software since 1979 and
has developed SAS procedures since 1984.

In addition to writing the companion book,Advanced ODS Graphics Examples, Warren has written
the following SAS/STAT documentation chapters: Chapter 20, “Using the Output Delivery System”
(SAS/STAT User's Guide), Chapter 21, “Statistical Graphics Using ODS” (SAS/STAT User's Guide),
Chapter 22, “ODS Graphics Template Modi�cation” (SAS/STAT User's Guide), and Chapter 23,
“Customizing the Kaplan-Meier Survival Plot” (SAS/STAT User's Guide).

Warren maintains 11 SAS/STAT procedures. He has developed 20 SAS macros for experimental
designs for linear and choice models. His 1,300-page book about discrete choice and other marketing
research methods,Marketing Research Methods in SAS, is available free on the web:http://
support.sas.com/resources/papers/tnote/tnote_marketresearch.html . War-
ren has also developed the world's largest collection of orthogonal arrays:http://support.

This book provides an introduction to the statistical graphics (SG) procedure SGPLOT and the Graph
Template Language (GTL). Creating a graph by using PROC SGPLOT is a one-step process. You
call PROC SGPLOT and provide one or more additional statements that describe the type of plot, the
variables, and so on. Then ODS Graphics creates the plot, and you look at it in one or more SAS
destinations as you would any other SAS output. Creating a graph by using the GTL is a two-step
process. You use PROC TEMPLATE to de�ne a graph template that includes multiple statements
that describe the type of plot, the variables, and so on. You call PROC SGRENDER, specifying
the template and the data. Then ODS Graphics creates the plot, and you look at the results. Most
examples show you how to make a graph with PROC SGPLOT and then show you how to make the
same graph by using PROC TEMPLATE, the GTL, and PROC SGRENDER. Some examples also
show statistical procedures that make the same type of graph. If you can make the graph you need
by using PROC SGPLOT or a statistical procedure, those methods will usually be easier than using
the GTL. Throughout this book, both methods are shown together to help you understand simple
GTL examples, even when the GTL is not needed, so that you can use it to easily construct more
complicated graphs that are not available with PROC SGPLOT.

Sections are devoted to different types of plots, beginning with sections on the scatter plot and �t
plots. The �rst few examples provide much greater detail than the later examples, so you should
read the �rst few examples before moving on to subsequent examples. Speci�cally, the common
GTL statements (DEFINE, BEGINGRAPH, LAYOUT OVERLAY, and so on) are explained in detail
only in the section “Scatter Plot” on page 3. Subsequent examples assume that you have read this
�rst example. You should also read the section “Regression Fit Plot” on page 6 before proceeding
to other examples. Other than those sections, you do not need to read the sections in order. If you
come across statements that are unfamiliar, you can use the index to �nd the �rst example of them.
The earlier examples go into more detail about the basics, and the later examples reveal more of the
nuances and power of the GTL and the SG procedures.

You can click on each of the following PROC SGPLOT statements or go to the page listed to see the
example that introduces that statement:

BAND on page 23 HISTOGRAM on page 73 SYMBOLCHAR on page 45
BLOCK on page 128 HLINE on page 111 SYMBOLIMAGE on page 45
BUBBLE on page 109 INSET on page 32 TEXT on page 109
DENSITY on page 78 KEYLEGEND on page 23 VBAR on page 49
DOT on page 98 LINEPARM on page 16 VBARBASIC on page 49
DROPLINE on page 84 LOESS on page 11 VBARPARM on page 49
ELLIPSE on page 104 NEEDLE on page 114 VBOX on page 88
FRINGE on page 82 PBSPLINE on page 14 VECTOR on page 117
GRADLEGEND on page 37 POLYGON on page 100 VLINE on page 111
HBAR on page 49 REFLINE on page 117 WATERFALL on page 131
HBARBASIC on page 49 REG on page 6 X2AXIS on page 57
HBARPARM on page 49 SCATTER on page 3 XAXIS on page 3
HBOX on page 88 SERIES on page 95 XAXISTABLE on page 64
HEATMAP on page 37 SPLINE on page 95 Y2AXIS on page 57
HEATMAPPARM on page 37 STEP on page 115 YAXIS on page 3
HIGHLOW on page 49 STYLEATTRS on page 16 YAXISTABLE on page 64

1.1 Scatter Plot

(Code links work in Adobe Reader and Internet Explorer.)

A scatter plot is a graphical display of two quantitative variables that uses Cartesian coordinates. The
data are displayed as points, each having the value of one variable on the horizontal axis and the
value of the other variable on the vertical axis.

You can use PROC SGPLOT to make a scatter plot as follows:

proc sgplot data=sashelp.class;
title �Weight by Height�;
scatter y=weight x=height;

run;

The input SAS data set is speci�ed in the DATA= option, and the title is speci�ed in the TITLE
statement. The SCATTER statement constructs a plot that has the variableWeight on the Y axis and
the variableHeight on the X axis. The results are displayed in Figure 1.1.

Figure 1.1 Simple Scatter Plot

Alternatively, you can use PROC TEMPLATE and the GTL. The following step creates a graph
template that can be used to create a scatter plot of theSashelp.Class data set:

proc template;
define statgraph classscatter;

begingraph;
entrytitle �Weight by Height�;
layout overlay;

scatterplot y=weight x=height;
endlayout;

endgraph;
end;

run;

The step consists of a PROC TEMPLATE statement followed by three or more nested blocks
of statements: DEFINE/END, BEGINGRAPH/ENDGRAPH, and LAYOUT/ENDLAYOUT. The
outermost layer of a graph template begins with a DEFINE STATGRAPH statement and ends with
an END statement. The �nal RUN statement is not required. Each template is compiled when SAS
encounters the END statement that matches the DEFINE statement. This template, like all graph
templates, begins with a DEFINE STATGRAPH statement followed by a template name. This step
creates a graph template calledClassScatter . The BEGINGRAPH statement is next. It provides a
place for you to specify options that affect the graph size, the graph border, and the graph background
color. Most statements that construct the graph are speci�ed inside the BEGINGRAPH/ENDGRAPH
block. Exceptions include the DYNAMIC, MVAR, and NMVAR statements, which provide names
for dynamic and macro variables that control details of how the graph is created. The graph title is
speci�ed in the ENTRYTITLE statement that follows the BEGINGRAPH statement. System titles
and footnotes (speci�ed in TITLEn and FOOTNOTEn statements) do not appear in the graph.

For a single graph such as this one (in contrast to a panel that contains two or more graphs), the
next statement is usually a LAYOUT OVERLAY statement. It is accompanied by an ENDLAYOUT
statement, and the statements that create the graph are in between. The LAYOUT OVERLAY
statement provides a place for you to specify options that control the ticks, tick labels, axes, axis
type, axis labels, grids, and so on. In this case, no options are speci�ed.

The only graph statement is a SCATTERPLOT statement that plots the variableWeight on the Y axis
and the variableHeight on the X axis. You can use this template to create the scatter plot by using
the following step:

proc sgrender data=sashelp.class template=classscatter;
run;

This PROC SGRENDER step consists of a DATA= option and a TEMPLATE= option. Some PROC
SGRENDER steps are more involved, but most are this simple. All the other instructions are in the
template. The results match those displayed in Figure 1.1.

The following step changes the axes and speci�es other options:

proc sgplot data=sashelp.class;
title �Weight by Height�;
scatter y=weight x=height / datalabel=name

markerattrs=(symbol=circlefilled
color=black size=3px);

xaxis offsetmin=0.05 offsetmax=0.05 label=�Class Height�;
yaxis offsetmin=0.05 offsetmax=0.05 label=�Class Weight�

values=(50 to 150 by 25);
run;

In PROC SGPLOT, the X- and Y-axis options are speci�ed in the XAXIS and YAXIS statements,
respectively. Offsets are speci�ed so that the �rst 5% (OFFSETMIN=0.05) and the last 5% (OFF-
SETMAX=0.05) of each axis are left blank. Aesthetically, it is often good to have a small amount
of separation between the axes and the extreme points in the plot. The axis label is speci�ed
by using the LABEL= option. The VALUES= option speci�es the Y-axis ticks. The DATALA-
BEL=Name option in the SCATTER statement uses theName variable to label the points. The
markerattrs=(symbol=circlefilled color=black size=3px) option speci�es black �lled
circle markers, three pixels in size. The results are displayed in Figure 1.2.

Figure 1.2 Scatter Plot of Labeled Points

The following steps make a graph that is identical to the graph in Figure 1.2:

proc template;
define statgraph classscatter;

begingraph;
entrytitle �Weight by Height�;
layout overlay /

xaxisopts=(offsetmin=0.05 offsetmax=0.05 label=�Class Height�)
yaxisopts=(offsetmin=0.05 offsetmax=0.05 label=�Class Weight�

linearopts=(tickvaluesequence=(start=50
end=150 increment=25) viewmin=50));

scatterplot y=weight x=height / datalabel=name
markerattrs=(symbol=circlefilled

color=black size=3px);
endlayout;

endgraph;
end;

run;

proc sgrender data=sashelp.class template=classscatter;
run;

In the GTL, X- and Y-axis options are speci�ed in the LAYOUT OVERLAY statement in
the XAXISOPTS= and YAXISOPTS= options, respectively. The tick values are speci�ed us-
ing these options:linearopts=(tickvaluesequence=(start=50 end=150 increment=25)
viewmin=50)) . The LINEAROPTS= option is used for linear-axis options. Other axis types (log,
discrete, and time) have different options. This set of options produces ticks from 50 to 150 by an
increment of 25. However, ODS Graphics does not automatically use all the ticks if they are too far
outside the range of the data. The VIEWMIN=50 option ensures that the smallest tick is displayed.
Without this option, the �rst tick is not displayed for these data. There is also a VIEWMAX= option
that is not used in this example.

The next section starts with a scatter plot and adds regression �t functions to it.

1.2 Regression Fit Plot

A regression �t plot consists of a scatter plot of two quantitative variables along with an overlaid
linear or nonlinear �t function. The Y coordinates of the �t function are often computed by using the
method of least squares, although other methods are illustrated in later examples. You can use PROC
SGPLOT to make a linear �t plot as follows:

proc sgplot data=sashelp.class noautolegend;
title �Linear Regression�;
reg y=weight x=height;

run;

The results are displayed in Figure 1.3. The REG statement plots the variableWeight on the Y
axis and the variableHeight on the X axis. A single statement produces both the scatter plot and
the �t line. The NOAUTOLEGEND statement in PROC SGPLOT suppresses the display of the

automatically generated legend. In a simple �t plot such as this, you do not need a legend to explain
that the graph contains a scatter plot and a �t function.

The REG statement in PROC SGPLOT automatically speci�es subpixel rendering (SUBPIXEL=ON).
This option produces a smoother and less jagged �t function, at a cost of slower run times. For an
example such as this, the extra run time is negligible. The following statement enables subpixel
rendering for all subsequent examples (so that you do not need to specify it in any of the templates):

ods graphics on / subpixel=on;

Figure 1.3 Linear Fit Function

The following step creates a template that can be used to create a scatter plot of theSashelp.Class
data set along with a linear �t:

proc template;
define statgraph classreg;

begingraph;
entrytitle �Linear Regression�;
layout overlay;

scatterplot y=weight x=height;
regressionplot y=weight x=height;

endlayout;
endgraph;

end;
run;

The SCATTERPLOT statement plots the variableWeight on the Y axis and the variableHeight on the
X axis. The REGRESSIONPLOT statement �ts a line through the scatter plot that is de�ned by the
variableWeight on the Y axis and the variableHeight on the X axis. The �t function is linear, because
the default degree is 1. You can use this template and create the plot by using PROC SGRENDER:

proc sgrender data=sashelp.class template=classreg;
run;

The results match those displayed in Figure 1.3.

You can specify the DEGREE=3 option for a cubic polynomial �t function through the points that
are de�ned by the variables. You can specify DEGREE=2 for a quadratic �t function. The following
steps use PROC SGPLOT and the GTL to create a cubic �t function:

proc sgplot data=sashelp.class noautolegend;
title �Cubic Polynomial Model�;
reg y=weight x=height / degree=3;

run;

proc template;
define statgraph classreg;

begingraph;
entrytitle �Cubic Polynomial Model�;
layout overlay;

scatterplot y=weight x=height;
regressionplot y=weight x=height / degree=3;

endlayout;
endgraph;

end;
run;

proc sgrender data=sashelp.class template=classreg;
run;

The graphs are identical; one is displayed in Figure 1.4.

Figure 1.4 Cubic Fit Function

The SG procedures work by writing a template in the GTL and using it to produce a graph. PROC
SGPLOT has an option in the PROC statement, the TMPLOUT= option, that writes the generated
template to a �le. You can look at that template, use it with PROC SGRENDER, or modify it �rst
and then use it with PROC SGRENDER.

The following step illustrates this option:

proc sgplot data=sashelp.class noautolegend tmplout=�fittmplt.sas�;
title �Cubic Fit Function�;
reg y=weight x=height / degree=3;

run;

The generated template is displayed in Figure 1.5, and it closely matches the template that is used in
this example. The generated template usually speci�es some options that you do not need in order to
re-create the graph.

Figure 1.5 Fit Plot Generated Template

proc template;
define statgraph sgplot;
begingraph / collation=binary subpixel=on;
EntryTitle "Cubic Fit Function" /;
layout overlay / yaxisopts=(labelFitPolicy=Split) y2axisopts=(labelFitPolicy=Split);

ScatterPlot X=Height Y=Weight / primary=true;
RegressionPlot X=Height Y=Weight / NAME="REG" LegendLabel="Regression" Degree=3;

endlayout;
endgraph;
end;
run;

You can use SAS/STAT procedures to create �t plots when you want more explicit control over the
model or more detailed output. Fit plots are created directly by several procedures, including the
REG, GLM, and TRANSREG procedures. The following steps use PROC REG and PROC GLM to
create �t plots. The results are displayed in Figure 1.6 and Figure 1.7.

proc reg data=sashelp.class;
model weight = height;

quit;

proc glm data=sashelp.class;
model weight = height;

quit;

Figure 1.6 PROC REG Fit Plot

Figure 1.7 PROC GLM Fit Plot

1.2.1 Loess Fit Plot

A loess �t plot consists of a scatter plot of two quantitative variables along with an overlaid nonlinear
�t function. The loess �t function is found by using a nonparametric locally weighted scatter plot
smoothing technique (Cleveland, Devlin, and Grosse 1988). The following steps show two ways to
display a loess �t in a scatter plot:

proc sgplot data=sashelp.enso noautolegend;
title �Loess Fit Function�;
loess y=Pressure x=Month;

run;

proc template;
define statgraph loess;

begingraph;
entrytitle �Loess Fit Function�;
layout overlay;

scatterplot y=Pressure x=Month;
loessplot y=Pressure x=Month;

endlayout;
endgraph;

end;
run;

proc sgrender data=sashelp.enso template=loess;
run;

The El Niño Southern Oscillation (ENSO) data set contains measurements of monthly averaged
atmospheric pressure differences between Easter Island and Darwin, Australia, for a period of 168
months (National Institute of Standards and Technology 1998). The ENSO data set is particularly
well suited for illustrating nonlinear �t functions. The data show both seasonal variations and
variations caused by El Niño. The ENSO data set is available in theSashelp library.

The LOESS statement in PROC SGPLOT and the LOESSPLOT statement in the GTL �t a non-
parametric regression function through the points that are de�ned by the variablePressure on the
Y axis and the variableMonth on the X axis. The NOAUTOLEGEND statement is used in PROC
SGPLOT to suppress the display of the automatically generated legend. The graphs are identical;
one is displayed in Figure 1.8.

Figure 1.8 Loess Fit Plot

You can use PROC LOESS to create a loess �t plot when you want more explicit control of the
results and when you are interested in a detailed statistical analysis of your data rather than simply a
graph with a �t function. You can compute a loess �t for these data in two different ways by using
the following steps:

proc loess data=sashelp.enso;
model Pressure=Month;

run;

proc loess data=sashelp.enso;
model Pressure=Month / select=AICC(global);

run;

The results of these steps are displayed in Figure 1.9 and Figure 1.10. The plot in Figure 1.9 displays
a local optimum (El Niño cycle), and the plot in Figure 1.10 displays the global optimum (seasonal
variation). PROC LOESS �nds the local optimum by default and needs some nudging to �nd the
global optimum. For more information about loess and this example, see Chapter 71, “The LOESS
Procedure” (SAS/STAT User's Guide).

Figure 1.9 Locally Optimal Loess Fit

Figure 1.10 Globally Optimal Loess Fit

1.2.2 Penalized B-Spline Fit Plot

A penalized B-spline �t plot consists of a scatter plot of two quantitative variables along with an
overlaid nonlinear �t function. The penalized B-spline �t function is found by using a �exible
method (Eilers and Marx 1996) that automatically chooses the smoothing parameter that minimizes
the corrected AIC criterion (AICC). The following steps show two ways to make a penalized B-spline
�t plot:

proc sgplot data=sashelp.enso noautolegend;
title �Penalized B-Spline Fit Function�;
pbspline y=Pressure x=Month;

run;

proc template;
define statgraph pbs;

begingraph;
entrytitle �Penalized B-Spline Fit Function�;
layout overlay;

scatterplot y=Pressure x=Month;
pbsplineplot y=Pressure x=Month;

endlayout;
endgraph;

end;
run;

proc sgrender data=sashelp.enso template=pbs;
run;

The PBSPLINE statement in PROC SGPLOT and the PBSPLINEPLOT statement in the GTL �t
a smooth function with an automatically chosen smoothing parameter through the points that are
de�ned by the variablePressure on the Y axis and the variableMonth on the X axis. The graphs are
identical; one is displayed in Figure 1.11.

Figure 1.11 Penalized B-Spline Fit Plot

You can use PROC TRANSREG to create a penalized B-spline �t plot when you want more explicit
control over the results and when you are interested in a detailed statistical analysis of your data
rather than simply a graph with a �t function. You can compute a penalized B-spline �t for these
data in two different ways by using the following steps:

proc transreg data=sashelp.enso;
model identity(pressure) = pbspline(month);

run;

proc transreg data=sashelp.enso;
model identity(pressure) = pbspline(month / sbc lambda=2 10000 range);

run;

The results of these steps are displayed in Figure 1.12 and Figure 1.13. The plot in Figure 1.12
displays the global optimum (seasonal variation), and the plot in Figure 1.13 displays a local optimum
(El Niño cycle). PROC TRANSREG �nds the global optimum by default and needs some nudging
away from small smoothing parameters to �nd the local optimum. For more information about
penalized B-splines and this example, see Chapter 117, “The TRANSREG Procedure” (SAS/STAT
User's Guide).

Figure 1.12 Globally Optimal Penalized B-Spline Fit

Figure 1.13 Locally Optimal Penalized B-Spline Fit

1.2.3 Grouped Regression Fit Plot

Previous examples showed several types of �t plots that have a single quantitative dependent variable
and a single quantitative independent variable. This example uses a model that also has a classi�cation
or group variable, and �ts a model that has separate intercepts and functions for each group. The
following steps show two ways to produce a graph, each displaying a separate cubic �t function for
males and for females:

proc sort data=sashelp.class out=class;
by descending sex;

run;

proc sgplot data=class;
title �Separate Fit by Sex�;
reg y=weight x=height / group=sex degree=3;

run;

proc template;
define statgraph classgroup;

begingraph;
entrytitle �Separate Fit by Sex�;
layout overlay;

scatterplot y=weight x=height / group=sex;
regressionplot y=weight x=height / group=sex degree=3

name=�reg�;
discretelegend �reg� / title=�Sex�;

endlayout;
endgraph;

end;
run;

proc sgrender data=class template=classgroup;
run;

The graphs are identical; one is displayed in Figure 1.14.

Figure 1.14 Regression with a Group Variable

Because of the GROUP= variable in the REG and REGRESSIONPLOT statements, separate compu-
tations are performed for each group. Because of the GROUP= variable in the REG, SCATTERPLOT,
and REGRESSIONPLOT statements, the two groups of observations are displayed differently ac-
cording to the rules speci�ed in the ODS style. With the HTMLBlue style, which is the default style
in this book, members of the �rst group (males) are displayed as blue circles with a solid blue �t
function, and members of the second group (females) are displayed as red circles with a red solid
�t function. Males are sorted ahead of females because of the DESCENDING option in the BY
statement in PROC SORT.

The GTL and PROC SGRENDER require the input data set to be sorted by the group variable.
In PROC SGPLOT, the GROUP= option in the REG statement creates the proper subsets of data
without requiring you to sort �rst. Both methods use the sorted data in this example to ensure the
same grouping for both methods.

The REGRESSIONPLOT statement is given a name with the NAME=�reg� option. That name is
speci�ed in the DISCRETELEGEND statement to produce the legend. A different name could also
have been provided in the SCATTERPLOT statement, and that name could also have been speci�ed
in the DISCRETELEGEND statement. However, with a �t plot, the �t functions are the same colors
as their corresponding points, so the legend for the REGRESSIONPLOT statement is suf�cient. The
TITLE= option in the DISCRETELEGEND statement provides a title for the legend.

This example shows how you can change the group colors:

proc sort data=sashelp.class out=class;
by sex;

run;

proc sgplot data=class;
title �Separate Fit by Sex�;
styleattrs datacontrastcolors=(pink blue) datasymbols=(circlefilled);
reg y=weight x=height / group=sex degree=3;

run;

proc template;
define statgraph classgroup;

begingraph / datacontrastcolors=(pink blue) datasymbols=(circlefilled);
entrytitle �Separate Fit by Sex�;
layout overlay;

scatterplot y=weight x=height / group=sex;
regressionplot y=weight x=height / group=sex degree=3

name=�reg�;
discretelegend �reg� / title=�Sex�;

endlayout;
endgraph;

end;
run;

proc sgrender data=class template=classgroup;
run;

The graphs are identical; one is displayed in Figure 1.15. The data are sorted by ascending values of
the variableSex. Colors are speci�ed in the STYLEATTRS statement in PROC SGPLOT and in the
BEGINGRAPH statement in the GTL. The color pink is used for the �rst group (females), and the
color blue is used for the second group (males). The markers are changed to �lled circles because
pink (which is a light color) tends to fade into the background when it is used for open circles.

Figure 1.15 Controlling the Appearance with a Group Variable

In the previous graphs, the same marker was used for males and females. This example shows how
you can vary the markers:

ods graphics on / attrpriority=none subpixel=on;

proc sort data=sashelp.class out=class;
by sex;

run;
proc sgplot data=class;

title �Separate Fit by Sex�;
styleattrs datacontrastcolors=(pink blue)

datalinepatterns=(solid)
datasymbols=(diamondfilled circlefilled);

reg y=weight x=height / group=sex degree=3;
run;

proc template;
define statgraph classgroup;

begingraph / datacontrastcolors=(pink blue)
datalinepatterns=(solid)
datasymbols=(diamondfilled circlefilled);

entrytitle �Separate Fit by Sex�;
layout overlay;

scatterplot y=weight x=height / group=sex;
regressionplot y=weight x=height / group=sex degree=3

name=�reg�;
discretelegend �reg� / title=�Sex�;

endlayout;
endgraph;

end;
run;

proc sgrender data=class template=classgroup;
run;

ods graphics on / reset=all subpixel=on;

The graphs are identical; one is displayed in Figure 1.16.

Figure 1.16 Varying Markers by Using a Group Variable

Styles have varying attribute priorities. An ATTRPRIORITY=NONE style uses a different combina-
tion of colors, markers, and line patterns to distinguish every group. An ATTRPRIORITY=COLOR
style uses only colors to distinguish the �rst 12 groups. Most styles (such as Default, Statistical,
and Analysis) are ATTRPRIORITY=NONE styles. A few styles (such as HTMLBlue, Pearl, and
Sapphire) are ATTRPRIORITY=COLOR styles. You can specify the ATTRPRIORITY= option
in the ODS GRAPHICS statement to change the attribute priority of any style. If the ATTRPRI-
ORITY=NONE option had not been speci�ed in the ODS GRAPHICS statement for this style
(HTMLBlue), the �rst marker (the �lled diamond) would have been used for both groups.

When you use the ATTRPRIORITY=NONE option, the �rst line is solid and the second line is
dashed by default. To vary markers and colors but not line patterns, specify a single line pattern. To
vary line patterns and colors but not markers (not shown), specify a single marker.

When you have grouped data, the combination of ATTRPRIORITY=NONE and DATALINEPAT-
TERNS=(SOLID) usually produces nice graphs:

ods graphics on / attrpriority=none subpixel=on;

proc sgplot data=class;
styleattrs datalinepatterns=(solid);
. . .

run;

Points on the graph are more easily distinguished when markers vary, but varying line patterns are
often not necessary in order to distinguish the different functions.

If you want your regression line to extend the length of the plot, you can use the LINEPARM
statement. The following steps create the graphs in Figure 1.17 and Figure 1.18:

ods graphics on / attrpriority=none;
title;

proc reg data=sashelp.iris noprint
outest=s(rename=(petallength=Slope species=s)

keep=petallength int: species);
model sepallength = petallength;
by species;

quit;

data iris;
merge sashelp.iris s;

run;

proc sgplot;
title �Separate Fit by Sex�;
styleattrs datasymbols=(circlefilled squarefilled trianglefilled)

datalinepatterns=(solid);
lineparm x=0 y=intercept slope=slope / group=s lineattrs=(thickness=2)

nomissinggroup;
scatter y=sepallength x=petallength / group=species;
xaxis min=10;

run;

proc sgplot;
title �Separate Fit by Sex�;
styleattrs datasymbols=(circlefilled squarefilled trianglefilled)

datalinepatterns=(solid);
reg y=sepallength x=petallength / group=species;

run;

ods graphics on / reset=attrpriority;

The results in Figure 1.17 are created by PROC REG and are displayed by the LINEPARM statement.
The results in Figure 1.18 are directly computed and displayed by the REG statement. The MERGE
statement merges a data set that has 150 observations (Sashelp.Iris) and a data set that has 3
observations (S). TheIntercept, Slope, andS variables have 147 missing values in the merged data
set. The LINEPARM statement uses the 3 observations that have a nonmissing slope and intercept to
make the three regression lines. The NOMISSINGGROUP option ensures that a fourth group is not
created for missing values ofS.

Figure 1.17 Regression Lines Produced by the LINEPARM Statement

Figure 1.18 Regression Lines Produced by the REG Statement

1.2.4 Regression Fit Plot with Con�dence and Prediction Limits

This example introduces band plots and series plots. Band plots display con�dence limits as a �lled
band. A series plot is a series of line segments that connect consecutive points. You can use series
plots to display predicted values, con�dence limits, prediction limits, and so on. The following steps
illustrate how much SAS/STAT procedures that create ODS Graphics automatically do for you, how
much PROC SGPLOT automatically does for you, and how much you have to do for yourself if
you want to create the same or similar plots by explicitly specifying each of the graph components
in either PROC TEMPLATE or PROC SGPLOT. These explicit-speci�cation examples provide an
opportunity for you to understand statements and compare different ways of doing things; they do
not provide models for actual graph construction.

The following steps show two ways to create a cubic �t plot with con�dence and prediction limits:

proc transreg data=sashelp.class;
model identity(weight) = spline(height);
output out=class clm cli p;

run;

proc sgplot data=sashelp.class;
title �Cubic Fit�;
reg y=weight x=height / degree=3 clm cli;

run;

The graph in Figure 1.19 is created automatically by PROC TRANSREG when ODS Graphics
is enabled. Only the PROC and MODEL statements are required. The OUTPUT statement and
its options create an output data set that is used in subsequent steps, but they are not required in
order to make the �t plot. The graph in Figure 1.20 is created by PROC SGPLOT. The CLM
option (con�dence limits for mean predicted values) produces the inner band, which shows the
con�dence limits. The CLI option (con�dence limits for the individual predicted values) in the REG
statement produces the outer series plots (or equivalently, the outer un�lled band plot), which show
the prediction limits.

Figure 1.19 Cubic Fit Plot from PROC TRANSREG

Figure 1.20 Cubic Fit Plot from PROC SGPLOT

The two plots are similar but not identical. The titles are different, the legends are ordered differently,
and the �t function legend entries are labeled differently. In addition, PROC SGPLOT automatically
creates smoother functions by computing predicted values, con�dence limits, and prediction limits
for intermediate interpolated values. PROC TRANSREG, by default, computes these values for each

observed X value and does not add intermediate points. This difference is usually noticeable only for
small data sets such as the one used in this example.

The following steps use the output data set from PROC TRANSREG and produce the same plots by
manually specifying the scatter plot, �t function, con�dence limits, and prediction limits, each with a
separate statement:

proc template;
define statgraph classfit;

begingraph;
entrytitle �Cubic Fit�;
layout overlay;

bandplot limitupper=cmuweight limitlower=cmlweight x=height /
connectorder=axis outlineattrs=GraphConfidence
name=�95% Confidence Limits�;

bandplot limitupper=ciuweight limitlower=cilweight x=height /
connectorder=axis display=(outline)
outlineattrs=GraphPredictionLimits
name=�95% Prediction Limits�;

scatterplot y=weight x=height;
seriesplot y=pweight x=height / name=�Fit� legendlabel=�Fit�

connectorder=xaxis lineattrs=GraphFit;
discretelegend �Fit� �95% Confidence Limits�

�95% Prediction Limits�;
endlayout;

endgraph;
end;

run;

proc sgrender data=class template=classfit;
run;

proc sort data=class;
by height;

run;

proc sgplot data=class;
title �Cubic Fit�;
band upper=cmuweight lower=cmlweight x=height /

fillattrs=GraphConfidence
name=�b1� legendlabel=�95% Confidence Limits�;

band upper=ciuweight lower=cilweight x=height /
nofill lineattrs=GraphPredictionLimits
name=�b2� legendlabel=�95% Prediction Limits�;

scatter y=weight x=height;
series y=pweight x=height / legendlabel=�Fit� lineattrs=GraphFit name=�f�;
keylegend �f� �b1� �b2�;

run;

The following GTL statement produces the con�dence limits:

bandplot limitupper=cmuweight limitlower=cmlweight x=height /
connectorder=axis outlineattrs=GraphConfidence
name=�95% Confidence Limits�;

A �lled band is displayed across the range of the X-axis variableHeight between the lower-limit
variablecmlweight and the upper-limit variablecmuweight. The CONNECTORDER=AXIS option
connects the data points from left to right along the X axis. The OUTLINEATTRS= option speci�es
theGraphConfidence style element for the outline. The NAME= option provides a statement name
so that the con�dence limit information can be added to the legend.

The following GTL statement produces the prediction limits:

bandplot limitupper=ciuweight limitlower=cilweight x=height /
connectorder=axis display=(outline)
outlineattrs=GraphPredictionLimits
name=�95% Prediction Limits�;

The limits are displayed across the range of the X-axis variableHeight between the lower-limit
variablecilweight and the upper-limit variableciuweight. The CONNECTORDER=AXIS option
connects the data points from left to right along the X axis. The DISPLAY= option displays only
the outline of the band rather than the entire band. The OUTLINEATTRS= option speci�es the
GraphPredictionLimits style element for the outline. The NAME= option provides a statement
name so that the prediction limit information can be added to the legend.

The following GTL statement produces the scatter plot:

scatterplot y=weight x=height;

It plots the variableWeight on the Y axis and the variableHeight on the X axis.

The following GTL statement produces the series plot that displays the �t function:

seriesplot y=pweight x=height / name=�Fit� legendlabel=�Fit�
connectorder=xaxis lineattrs=GraphFit;

The predicted values were previously computed by PROC TRANSREG, and the SERIESPLOT
statement connects them to provide the �t function. The label for the legend is�Fit� . The
LINEATTRS= speci�es theGraphFit style for the �t line. As in the BANDPLOT statements, the
CONNECTORDER=AXIS and NAME= options are speci�ed.

The following GTL statement produces the legend from the three named statements:

discretelegend �Fit� �95% Confidence Limits� �95% Prediction Limits�;

The results are displayed in Figure 1.21.

Figure 1.21 Cubic Fit Plot Constructed by Using the GTL

PROC SGPLOT statements do not have a CONNECTORDER=AXIS option, so the data are sorted
by the X=Height variable before the graph is made. The �rst plotting statement in PROC SGPLOT is
as follows:

band upper=cmuweight lower=cmlweight x=height / fillattrs=GraphConfidence
name=�b1� legendlabel=�95% Confidence Limits�;

A �lled band is displayed across the range of the X-axis variableHeight between the lower-limit
variablecmlweight and the upper-limit variablecmuweight. The FILLATTRS= option speci�es the
GraphConfidence style element for the �ll (the band). The NAME= option provides a statement
name so that the con�dence limit information can be added to the legend. The LEGENDLABEL=
option provides the text that is displayed in the legend along with the con�dence limits information.

The following PROC SGPLOT statement produces the prediction limits:

band upper=ciuweight lower=cilweight x=height /
nofill lineattrs=GraphPredictionLimits
name=�b2� legendlabel=�95% Prediction Limits�;

The limits are displayed across the range of the X-axis variableHeight between the lower-limit
variablecilweight and the upper-limit variableciuweight. The NOFILL option produces an outline
instead of a band. The LINEATTRS= option speci�es theGraphPredictionLimits style element
for the outline. The NAME= option provides a statement name so that the prediction limit information
can be added to the legend. The LEGENDLABEL= option provides the text that is displayed in the
legend along with the prediction limits information.

The following PROC SGPLOT statement produces the scatter plot:

scatter y=weight x=height;

It plots the variableWeight on the Y axis and the variableHeight on the X axis.

The following PROC SGPLOT statement produces the series plot that displays the �t function:

series y=pweight x=height / legendlabel=�Fit� lineattrs=GraphFit name=�f�;

The predicted values were previously computed by PROC TRANSREG, and the SERIES statement
connects them to provide the �t function. The label for the legend is�Fit� . The LINEATTRS=
speci�es theGraphFit style for the �t line. As in the BAND statements, the NAME= option is
speci�ed.

The following PROC SGPLOT statement produces the legend from the three named statements:

keylegend �f� �b1� �b2�;

The results are identical to the graph displayed in Figure 1.21.

Both the REG statement in PROC SGPLOT and the REGPLOT statement in the GTL accept the
following options and many others:

� ALPHA=0.05, the default, produces 95% con�dence and prediction limits. Specify AL-
PHA=0.01 for 99% con�dence and prediction limits.

� CURVELABEL='string' speci�es a label for the regression function.

The previous steps used the GTL, PROC TEMPLATE, and PROC SGRENDER to display regression
results that were computed by another procedure (PROC TRANSREG). The next series of steps
does all the computations within the GTL. They illustrate the MODELBAND statement and also the
technique of using and overriding style elements to control the appearance of the con�dence and
prediction limits.

The following steps create the graph in Figure 1.22:

proc template;
define statgraph classfit1;

begingraph;
entrytitle �Cubic Fit�;
layout overlay;

modelband "cliband" / display=(outline);
modelband "clmband";
scatterplot y=weight x=height;
regressionplot y=weight x=height / degree=3

clm=�clmband� cli=�cliband�;
endlayout;

endgraph;
end;

run;

proc sgrender data=sashelp.class template=classfit1;
run;

The CLM=�clmband� option in the REGRESSIONPLOT statement displays the con�dence limits
according to the options speci�ed in the MODELBAND statement named�clmband� . This
MODELBAND statement has no options, so a default band plot, which consists of a �lled band
with no outline, is produced. Similarly, the CLI=�cliband� option displays the prediction limits
according to the options speci�ed in the MODELBAND statement named�cliband� . This
MODELBAND statement speci�es that an outline of the band is to be displayed, but the band is
not �lled. Statements are executed in the order in which they are speci�ed in the template. The
prediction limits are drawn �rst, then the con�dence limits, then the scatter plot, then the �t function.
Plot elements that are placed later can obscure previously placed plot elements. Hence, you do not
want to draw the band plots last, because you will not be able to see the �t function and many of the
points.

Figure 1.22 Prediction Limits Outlined

The following steps create the graph in Figure 1.23:

proc template;
define statgraph classfit2;

begingraph;
entrytitle �Cubic Fit�;
layout overlay;

modelband "cliband" / fillattrs=GraphConfidence;
modelband "clmband" / fillattrs=GraphConfidence2;
scatterplot y=weight x=height;
regressionplot y=weight x=height / degree=3

clm="clmband" cli="cliband";
endlayout;

endgraph;
end;

run;

proc sgrender data=sashelp.class template=classfit2;
run;

In this example, theFillAttrs=GraphConfidence option uses theGraphConfidence style ele-
ment to control the appearance of the prediction limits, and theFillAttrs=GraphConfidence2
option uses theGraphConfidence2 style element to control the appearance of the con�dence limits.

Figure 1.23 Con�dence Style Elements

The following steps create the graph in Figure 1.24:

proc template;
define statgraph classfit3;

begingraph;
entrytitle �Cubic Fit�;
layout overlay;

modelband "cliband" / outlineattrs=GraphPredictionLimits
display=(outline)
datatransparency=0.5;

modelband "clmband" / fillattrs=GraphConfidence
datatransparency=0.5;

scatterplot y=weight x=height;
regressionplot y=weight x=height / degree=3

clm="clmband" cli="cliband";
endlayout;

endgraph;
end;

run;

proc sgrender data=sashelp.class template=classfit3;
run;

The graph in Figure 1.24 has an appearance similar to that of the graph produced directly
by PROC TRANSREG (shown in Figure 1.19) and to the graph produced directly by PROC
SGPLOT (shown in Figure 1.20). The prediction limits are displayed as an outline with-
out �ll because of the DISPLAY=(OUTLINE) option. The style of the line is controlled by
theOutLineAttrs=GraphPredictionLimits option, which uses theGraphPredictionLimits
style element, which speci�es a dashed line in this style. The DATATRANSPARENCY=0.5 option is
speci�ed so that the bands and outlines are 50% as transparent as the default. The con�dence limits
are displayed as a band without an outline because of the default setting of the DISPLAY= option,
DISPLAY=(FILL). The style of the �ll is controlled by theFillAttrs=GraphConfidence option,
which uses theGraphConfidence style element.

Figure 1.24 Outline, Fill, and Transparency

The following steps create the graph in Figure 1.25:

proc template;
define statgraph classfit4;

begingraph;
entrytitle �Cubic Fit�;
layout overlay;

modelband "cliband" /
outlineattrs=GraphPredictionLimits(pattern=solid)
display=(outline)
datatransparency=0.5;

modelband "clmband" /
fillattrs=GraphConfidence(color=cx88AAAA)
datatransparency=0.5;

scatterplot y=weight x=height;
regressionplot y=weight x=height / degree=3

clm="clmband" cli="cliband";

endlayout;
endgraph;

end;
run;

proc sgrender data=sashelp.class template=classfit4;
run;

Figure 1.25 Style Element Overrides

These steps are similar to the preceding steps. However, one part of theGraphPredictionLimits
style element is overridden—namely, the line pattern. The optionOutLineAttrs=Graph-
PredictionLimits(Pattern=Solid) creates prediction limits by using all style elements
from the GraphPredictionLimits element except for the default line style. Similarly,
the FillAttrs=GraphConfidence(Color=cx88AAAA) option overrides the color of the
GraphConfidence style element. All colors can be speci�ed in values of the form CXrrggbb, where
the last six characters specify RGB (red, green, blue) values on the hexadecimal scale of 00 to FF (0
to 255 base 10). HLS (hue/light/saturation) color speci�cations and all other color speci�cations
available in SAS/GRAPH are also available in ODS Graphics.

1.2.5 Text Insets and Special Characters

This section shows how to insert text into a graph, including text with superscripts (such as R2) and
Greek letters with special characters (such asO� ). The graph is the cubic �t function from the section
“Regression Fit Plot” on page 6. The following step runs PROC TRANSREG with a cubic polynomial
model and creates an ODS output data set with regression �t statistics, including R-square and the
mean of the dependent variable:

proc transreg data=sashelp.class ss2;
ods output fitstatistics=fs;
model identity(weight) = spline(height);

run;

The following step stores the R-square in a macro variable,R2, and the mean in a macro variable,
Mean:

data _null_;
set fs;
if _n_ = 1 then call symputx(�R2�, put(value2, 4.2 ), �G�);
if _n_ = 2 then call symputx(�mean�, put(value1, best6.), �G�);

run;

The R-square value is stored with a format of 4.2, and the mean is stored with the BEST6. format.
Both are stored in the global macro symbol table. These values are inserted into plots as inset text
entries in the next steps.

The following steps create a plot with the R-square and the mean in the top left corner:

proc template;
define statgraph classreg1;

mvar r2 mean;
begingraph;

entrytitle �Cubic Fit Function�;
layout overlay;

entry halign=left �R� {sup �2�} � = � r2
", Mean = " mean / valign=top;

scatterplot y=weight x=height;
regressionplot y=weight x=height / degree=3;

endlayout;
endgraph;

end;
run;

proc sgrender data=sashelp.class template=classreg1;
run;

The MVAR statement speci�es the two macro variables,R2 andMean, so that they can be used in
other parts of the template. The ENTRY statement in this example adds to the top left of the plot a
single line of text that consists of an “R2”, an equal sign, the formatted R-square value, a comma
and spaces, “Mean =”, and the mean value. The inset text entry is created from a series of values in
the ENTRY statement, including constant text, special characters, and values that are retrieved from
macro variables. The syntax'R' {sup '2'} is the syntax for superscripts. (Similarly, the syntax
for subscripts is'x' {sub 'i'} .) Notice that the macro variables are speci�ed without ampersands.
Therefore, their values are not retrieved before PROC TEMPLATE compiles the template; rather,
the values are retrieved when PROC SGRENDER is run. This approach enables you to create the
template once and use it repeatedly for different analyses that have different values for the two
statistics. Also notice that the speci�cation{sup '2'} does not appear in quotation marks. If it
had appeared in quotes, then “{sup '2'}” would appear in the graph and not “2”. The results are
displayed in Figure 1.26.

Figure 1.26 Inset Text with the GTL

The following steps insert the same information into the plot, but with two differences:

proc template;
define statgraph classreg2;

mvar r2 mean;
begingraph;

entrytitle �Cubic Fit Function�;
layout overlay;

layout gridded / autoalign=(topright topleft
bottomright bottomleft);

entry halign=left �R� {sup �2�} � = � r2;
entry halign=left "( * ESC* ){unicode mu}( * ESC* ){unicode hat} = "

mean / textattrs=GraphValueText(
family=GraphUnicodeText:FontFamily);

endlayout;
scatterplot y=weight x=height;
regressionplot y=weight x=height / degree=3;

endlayout;
endgraph;

end;
run;

proc sgrender data=sashelp.class template=classreg2;
run;

The text is split onto two lines, and the mean is labeled by “O� ” instead of “Mean.” The results of
this step are displayed in Figure 1.27. The LAYOUT GRIDDED statement groups the two ENTRY
statements, enabling them to be moved and positioned as a group with one immediately following the
other. You can place any number of ENTRY statements between the LAYOUT and ENDLAYOUT
statements. The LAYOUT GRIDDED statement has several options. For example, you can specify

the options OPAQUE=TRUE, BORDER=TRUE, and BACKGROUNDCOLOR=style-reference |
color to create an opaque box with a background and color. In this example, the AUTOALIGN=
option speci�es a list of preferred positions. ODS Graphics places the entries in the �rst available
position. In this example, there are data in the top right, so the entries are placed in the top left.

Figure 1.27 Special Characters with the GTL

Both ENTRY statements specify the HALIGN=LEFT option to left-align both entries. The second
ENTRY statement speci�es “O� ” with the speci�cation( * ESC* ){unicode mu}( * ESC* ){unicode
hat} , which appears in quotation marks. Special characters are speci�ed as Unicode characters
with an escape sequence. The( * ESC* ) speci�cation is a �ag that indicates that a special character
follows. Guidelines for when you use the( * ESC* ) speci�cation are discussed later in this section.
For now, simply note that the{sup '2'} speci�cation is not in quotes and is not escaped, whereas
the Unicode speci�cations are in quotes and are escaped. The TEXTATTRS= option is speci�ed so
that a font that has the Unicode characters is used. Notice that extra white space was added to better
align the parts of each line.

Greek letters are speci�ed by{unicode letter } or {unicode letter_U} . Lowercase Greek letters
are speci�ed by name (for example,Alpha , Beta , Gamma). Uppercase Greek letters are speci�ed
by appending an_U to the name (for example,Alpha_U , Beta_U , Gamma_U). Additional keywords
includeBar , Bar2 , Hat , Tilde , andPrime . These keywords are part of the Unicode range that is
called combining (nonspacing) diacritical marks, which means that they are drawn superimposed
on the previous character rather than in the next position. These marks should always immediately
follow the character they are part of. Hence,{Unicode Mu}{Unicode Hat} (which is correct) is
different from{Unicode Hat}{Unicode Mu} (which is backward and incorrect) and also different
from {Unicode Mu} {Unicode Hat} (which includes a space and so is also incorrect). The
Unicode Consortium (http://unicode.org/ ) provides a page of character codes athttp:
//www.unicode.org/charts/charindex.html .

The following step uses PROC SGPLOT and produces a plot that is almost identical to the plot
displayed in Figure 1.27 (which was created in the previous steps by PROC TEMPLATE and PROC
SGRENDER):1

proc sgplot data=sashelp.class noautolegend;
title �Cubic Fit Function�;
inset "R( * ESC* ){sup �2�} = &r2"

"( * ESC* ){unicode mu}( * ESC* ){unicode hat} = &mean" /
position=topleft;

reg y=weight x=height / degree=3;
run;

The INSET statement in this example produces two entries because two quoted strings are speci�ed.
In the INSET statement (unlike the ENTRY statement in the GTL), each string forms a new line,
and each string must be fully quoted, even when special characters are provided. Each Unicode,
superscript, and subscript speci�cation must be escaped because each must appear inside a quoted
string. The escape sequence tells ODS Graphics that a special character is coming. Otherwise,
each of these speci�cations appears “as is” in the graph. Because PROC SGPLOT does not have
an MVAR statement as the GTL does, the macro variables appear inside the double quotes and
appear with an ampersand. Their values are resolved when PROC SGPLOT is run. Note that the
speci�cation"R( * ESC* ){sup '2'} = &r2" contains double outer quotes and single inner quotes.
Double quotes are required for the outer quotes so that the macro variables will resolve. Double
quotes could also be used for the inner quotes, but they have to be doubled so that they do not end
the string (for example,"R( * ESC* ){sup ""2""} = &r2" ). No TEXTATTRS= option is needed
in PROC SGPLOT, which automatically switches to the appropriate font when it encounters the
Unicode characters.

The �nal steps in this example use PROC TEMPLATE and PROC SGRENDER and produce the
plot without any escape characters. The following steps make a graph that is identical to the one in
Figure 1.27:

proc template;
define statgraph classreg3;

mvar r2 mean;
begingraph;

entrytitle �Cubic Fit Function�;
layout overlay;

layout gridded / autoalign=(topright topleft
bottomright bottomleft);

entry halign=left �R� {sup �2�} � = � r2;
entry halign=left {unicode mu}{unicode hat} � = � mean /

textattrs=GraphValueText(family=GraphUnicodeText:FontFamily);
endlayout;

scatterplot y=weight x=height;
regressionplot y=weight x=height / degree=3;

endlayout;
endgraph;

end;
run;

1Many graphs in this book are said to be “almost identical” to some other graph. In most cases, a small amount of
white space shifts parts of one of the graphs. Some differences are as small as a pixel.

